

Métodos de Amostragem de Populações de Animais Silvestres

Programa SAS para análise de uma amostra estratificada

2009

Parte I

```
OPTIONS PS=56 PAGENO=1 ;  
DATA A;  
INPUT ESTRATO NMAIH NMINH VARH MEDIAH;  
DUM=1;  
TH=NMAIH*MEDIAH;  
VARMEDH=(VARH/NMINH)*((NMAIH-NMINH)/NMAIH);  
VARTH=(NMAIH**2)*VARMEDH;  
DATALINES;  
1 200 10 400 15  
2 100 10 900 10  
3 400 10 400 22  
;;;
```

Parte II

```
TITLE2'*** Métodos de Amostragem de Populações de Animais Silvestres ***';  
TITLE4'*** Análise de uma amostra estratificada ***';  
ODS PDF FILE='F:\.....\Coloque um nome!!!!!!!.PDF';  
PROC SORT DATA=A;  
  BY DUM;  
PROC MEANS DATA=A NOPRINT;  
  BY DUM;  
VAR NMAIH NMINH;  
OUTPUT OUT=B SUM=NMAI NMIN;
```

Parte III

```
DATA C;  
  MERGE A B;  
  BY DUM;  
MEDST=(NMAIH*MEDIAH)/NMAI;  
VARMEDST= ((NMAIH/NMAI)**2)*VARMEDH;  
WH=NMINH/NMIN;  
CALC1=(NMAIH**2)*VARH/WH;  
CALC2=NMAIH*VARH;  
GL=NMIN-1;  
T=TINV(0.975,GL);  
ED=10;  
PROC MEANS DATA=C NOPRINT;  
VAR TH VARTH MEDST VARMEDST CALC1 CALC2 GL T ED  
NMAI;  
OUTPUT OUT=D MEAN= X1 X2 X3 X4 X5 X6 GL T ED NMAI  
SUM=TOTST VARTST MEDST VARMEDST CALC1 CALC2 X7 X8 X9  
X10;
```

Parte IV

```
DATA E;  
  SET D;  
DROP X1 X2 X3 X4 X5 X6 X7 X8 X9 X10;  
ICMED=T*SQRT(VARMEDST);  
ICTOT=T*SQRT(VARTST);  
ERROMED=ICMED/MEDST*100;  
ERROTOT=ICTOT/TOTST*100;  
NAST=CALC1/((((NMAI**2)*(MEDST**2)*  
  (ED**2))/((100**2)*(T**2)))+CALC2);
```



Cuidado com os
parênteses.

Parte V

```
PROC PRINT DATA=E LABEL SPLIT='*' NOOBS;  
VAR MEDST TOTST ICMED ICTOT ERROMED NAST;  
LABEL MEDST='Média da*População'  
      TOTST='Total da*População'  
      ICMED='Intervalo*de*Confiança*da Média'  
      ICTOT='Intervalo*de*Confiança*do Total'  
      ERROMED='Erro*da*Amostragem*(%)'  
      NAST='Intensidade*da*Amostragem';  
FORMAT MEDST ICMED ERROMED COMMAX8.2  
      TOTST ICTOT NAST COMMAX8.0;  
RUN;  
ODS PDF CLOSE;
```

Resultado

The SAS System

**** Métodos de Amostragem de Populações de Animais Silvestres ****

**** Análise de uma amostra estratificada ****

Média da População	Total da População	Intervalo de Confiança da Média	Intervalo de Confiança do Total	Erro da Amostragem (%)	Intensidade da Amostragem
18,29	12.800	8,55	5.987	46,78	370